Das Lotka-Volterra Modell für das Predator-Prey Problem

```
> with(plots): with(DEtools):
```

Die Anzahl der Beutetiere x(t) und die Anzahl der Raubtiere y(t) sind gekoppelt durch die Lotka-Volterra Differentialgleichungen

```
> unassign('a','b','c','d'):
  LV1 := diff(x(t),t) = (a-b*y(t))*x(t);
  LV2 := diff(y(t),t) = (c*x(t)-d)*y(t);
```

$$LV1 := \frac{d}{dt} x(t) = (a - b\, y(t))\, x(t)$$

$$LV2 := \frac{d}{dt} y(t) = (c\, x(t) - d)\, y(t) \tag{1}$$

Wir wählen Koeffizienten

```
> a:=1; b:=0.03; c:=0.01; d:=0.4;
```

$$a := 1$$

$$b := 0.03$$

$$c := 0.01$$

$$d := 0.4 \tag{2}$$

und zeichnen das zugehörige Vektorfeld:

```
> DEplot([LV1,LV2], [x,y], t=0..30, x=0..140, y=0..80,
    stepsize=0.2, obsrange=FALSE, color=brown, linecolor=blue,
    labels=["Beutetiere","Raubtiere"], labelfont=[Helvetica,14]);
```



Hier ist das Vektorfeld kombiniert mit einigen Lösungskurven:

```
> DEplot([LV1,LV2], [x,y], t=0..30, x=0..140, y=0..80,
    [[x(0)=15,y(0)=15],[x(0)=20,y(0)=20],[x(0)=25,y(0)=25],[x(0)=
30,y(0)=30]],
    stepsize=0.2, obsrange=FALSE, color=brown, linecolor=blue,
    labels=["Beutetiere","Raubtiere"], labelfont=[Helvetica,14]);
```



Hier ist der zeitliche Verlauf der beiden Populationen:

```
> preyplot := DEplot([LV1,LV2], [x,y], t=0..30, x=0..140, y=0..80,
     [[x(0)=15,y(0)=15]], stepsize=0.2, obsrange=FALSE,
     scene=[t,x], linecolor=COLOR(RGB,0,0.7,0)):
  predplot := DEplot([LV1,LV2], [x,y], t=0..30, x=0..140, y=0..80,
     [[x(0)=15,y(0)=15]], stepsize=0.2, obsrange=FALSE,
     scene=[t,y], linecolor=red):
  display([preyplot,predplot],
     labels=["Zeit","Anzahl"], labelfont=[Helvetica,14],
     title="Beutetiere (grün), Raubtiere (rot)", titlefont=
  [Helvetica,14]);
```



Ein 3-dimensionaler Plot der Kurve (x(t),y(t),t) fasst die Gesamtinformation zusammen. Durch Anklicken und Ziehen mit der Maus kann der Blickwinkel gedreht werden.

```
> DEplot3d([LV1,LV2], [x,y], t=0..30, x=0..140, y=0..80,
     [[x(0)=15,y(0)=15]], stepsize=0.2, obsrange=FALSE,
     scene=[x,y,t], axes=normal, linecolor=blue,
     labels=["Beutetiere","Raubtiere","Zeit"], labelfont=
  [Helvetica,14]);
```



Mit dem nächsten Befehl ändern wir die Parameter und aktualisieren die Differentialgleichungen, um alles neu berechnen zu können. Experimentieren Sie auch mit anderen Werten! Dabei müssen Sie unter Umständen die Intervallgrenzen in den obigen Befehlen anpassen.

```
> a:=1; b:=0.03; c:=0.02; d:=0.4;
> LV1 := diff(x(t),t) = (a-b*y(t))*x(t):
  LV2 := diff(y(t),t) = (c*x(t)-d)*y(t):
```

$$a := 1$$

$$b := 0.03$$

$$c := 0.02$$

$$d := 0.4 \tag{3}$$

Hier ist das Vektorfeld kombiniert mit einigen Lösungskurven:

```
> DEplot([LV1,LV2], [x,y], t=0..30, x=0..140, y=0..80,
     [[x(0)=15,y(0)=15],[x(0)=20,y(0)=20],[x(0)=25,y(0)=25],[x(0)=
  30,y(0)=30]],
     stepsize=0.2, obsrange=FALSE, color=brown, linecolor=blue,
     labels=["Beutetiere","Raubtiere"], labelfont=[Helvetica,14]);
```



Als letztes addieren wir zur zweiten Gleichung eine negative Konstante, welche den Verlust an Raubtieren durch Jagen darstellt. Experimentieren Sie mit diesem und anderen Werten, z.B. mit e:=2. Kann es passieren, dass die Raubtiere aussterben? Kann es passieren, dass beide Arten aussterben?

```
> a:=1; b:=0.03; c:=0.01; d:=0.4; e:=1;
> LV1 := diff(x(t),t) = (a-b*y(t))*x(t):
  LV2 := diff(y(t),t) = (c*x(t)-d)*y(t)-e:
```

$$a := 1$$

$$b := 0.03$$

$$c := 0.01$$

$$d := 0.4$$

$$e := 1 \qquad (4)$$

Hier ist das Vektorfeld kombiniert mit einigen Lösungskurven:

```
> DEplot([LV1,LV2], [x,y], t=0..30, x=0..140, y=0..80,
     [[x(0)=15,y(0)=15],[x(0)=20,y(0)=20],[x(0)=25,y(0)=25],[x(0)=
  30,y(0)=30]],
     stepsize=0.2, obsrange=FALSE, color=brown, linecolor=blue,
     labels=["Beutetiere","Raubtiere"], labelfont=[Helvetica,14]);
```



Raubtiere
Beutetiere
0
10
20
30
40
50
60
70
80
20
40
60
80
100
120
140